TOKY V GRAFU

MAXIMÁLNÍ TOK SÍTÍ, MINIMALIZACE NÁKLADŮ SPOJENÝCH S DANOU HODNOTOU TOKU, FIXNÍ NÁKLADY, PŘEPRAVNÍ (TRANSHIPMENT) PROBLÉM.

Graf je útvar, který je možno znázornit obrázkem v rovině pomocí bodů (**uzly** grafu) a spojnic mezi body (**hrany** grafu). Hrany v grafu mohou být **orientované** nebo **neorientované**.
Orientovaný graf obsahuje na rozdíl od neorientovaného grafu orientované hrany.
Cesta v grafu je posloupnost orientovaných hran, při které vždy následující hrany začínají v uzlu, v němž končí předcházející hrana.
Sled v grafu je posloupnost vrcholů taková, že mezi každými dvěma po sobě jdoucími je hrana.
Cyklus (uzavřená cesta) je taková cesta, která začíná a končí v témže uzlu.
Souvislý graf je graf, u kterého mezi všemi dvojicemi uzlů existuje alespoň jedna cesta.
Nesouvislý graf je graf, u kterého neexistuje alespoň jedna cesta mezi všemi dvojicemi uzlů.
Úplný graf je takový graf, ve kterém je každá dvojice uzlů spojena hranou.
Strom je takový graf, který neobsahuje žádný cyklus.
Podgraf původního grafu je graf, který vznikne tím, že vynecháme z grafu některé uzly a příslušné hrany těchto uzlů.
Síť je graf, který je konečný, souvislý, orientovaný, acyklický a ohodnocený, v němž existuje jeden konečný a jeden počáteční uzel.



ZDROJ: WIKIPEDIA.ORG

HLEDÁNÍ MAXIMÁLNÍHO TOKU

Máme graf s jedním počátečním uzlem (zdroj) a jedním koncovým uzlem (místo určení). Známe kapacitu hran k_{ij} mezi každou dvojicí uzlů i, j. Naším cílem je maximalizovat celkový tok tímto grafem. Stačí tedy maximalizovat to, co celkem vyteče z prvního uzlu, nebo to, co celkem přiteče do posledního uzlu, protože to odpovídá celkovému toku grafem.

Zavedeme jednu proměnnou x_{ij}, která označuje objem toku z uzlu i do uzlu j.

$z = \sum_{j(1,j)\epsilon H} x_{1j} = \sum_{j(j,n)\epsilon H} x_{jn} \rightarrow max$	*Snažíme se maximalizovat buď celkový tok z prvního uzlu, nebo celkový tok do posledního uzlu. Sčítáme přes všechny hrany, které vedou z prvního uzlu, resp. do posledního uzlu.*
$\sum_{j(i,j)\epsilon H} x_{ij} - \sum_{k(k,i)\epsilon H} x_{ki} = 0 \quad i \neq 1,n$	*(1) Pro každý uzel kromě prvního a posledního musí platit, že to, co do i-tého uzlu celkem vteče, se musí rovnat tomu, co z něj celkem vyteče. Sčítáme přes všechny hrany, které vedou z daného uzlu a do něj.*
$0 \leq x_{ij} \leq k_{ij} \quad (i,j)\,\epsilon H$	*(2) Pro každou hranu mezi dvojicí uzlů i, j musí platit, že jí neproteče více, než kolik je její kapacita.*

```
Uvazujme orientovany graf s 6 uzly (a az f). V nasledujici tabulce jsou uvedeny
kapacity existujících hran. Jaky je maximalni tok timto grafem?
```

Spojení	Kapacita	Spojení	Kapacita
a,b	2	c,f	4
a,c	3	d,c	5
a,d	4	d,e	9
b,d	1	d,f	8
b,e	4	e,f	3

```
;
model:
sets:
uzel/a,b,c,d,e,f/;
tok(uzel,uzel)/a b,a c,a d,b d,b e,c f,d c,d e,d f,e f/: x,kapacita;   ! výčet hran;
endsets

data:
kapacita = 2 3 4 1 4 4 5 9 8 3;
enddata

max = @sum(tok(i,j)|i#EQ#1: x(i,j)); !kdybychom nepoužili výčet hran, pak podminku
zapíšeme jako @sum(uzel(j): x(1,j), ale pokud zadáváme hrany výčtem, pak musíme v
podmínkách pracovat s tou množinou, pro niž je tento výčet zadán;

@for(uzel(i)|i#NE#1#AND#i#NE#6: @sum(tok(i,j): x(i,j)) = @sum(tok(j,i): x(j,i)));
@for(tok: x<=kapacita);
end
```

Zadání



Řešení




```
Celkovy tok je 9.
```

HLEDÁNÍ NEJLEVNĚJŠÍHO TOKU

Máme graf s jedním počátečním uzlem (zdroj) a jedním koncovým uzlem (místo určení). Známe kapacitu hran k_{ij} mezi každou dvojicí uzlů i, j. Zároveň známe náklady c_{ij} spojené s jednotkou toku hranou *i*, *j*. Naším cílem je zajistit požadovanou hodnotu celkového toku T_0 s co nejnižšími náklady.

Zavedeme jednu proměnnou x_{ij}, která označuje objem toku z uzlu *i* do uzlu *j*.

$z = \sum_i \sum_{j_{(i,j)\epsilon H}} c_{ij} x_{ij} \rightarrow min$	*Snažíme se minimalizovat celkové náklady spojené s tokem.*
$\sum_{j_{(i,j)\epsilon H}} x_{ij} - \sum_{k_{(k,i)\epsilon H}} x_{ki} = 0 \quad i \neq 1, n$	*(1) Pro každý uzel kromě prvního a posledního musí platit, že to, co do i-tého uzlu celkem vteče, se musí rovnat tomu, co z něj celkem vyteče. Sčítáme přes všechny hrany, které vedou z daného uzlu a do něj.*
$0 \leq x_{ij} \leq k_{ij} \quad (i,j)\,\epsilon H$	*(2) Pro každou hranu mezi dvojicí uzlů i, j musí platit, že jí neproteče více, než kolik je její kapacita.*
$\sum_{j_{(1,j)\epsilon H}} x_{1j} \geq T_0$	*(3) To, co dohromady vyteče z prvního uzlu (tedy zároveň i celkový průtok grafem) musí být větší nebo rovno požadované hodnotě celkového toku,*

```
Uvazujme graf z předchozího prikladu. Navíc zname pro kazdou hranu naklady na
jednotku toku. Chceme, aby celkovy tok byl alespoň 9. Jake budou minimalni naklady?
```

Spojení	Kapacita	Naklady	Spojení	Kapacita	Naklady
a,b	2	5	c,f	4	8
a,c	3	4	d,c	5	2
a,d	4	6	d,e	9	1
b,d	1	6	d,f	8	6
b,e	4	7	e,f	3	2

```
;
model:
sets:
uzel/a,b,c,d,e,f/;
tok(uzel,uzel)/a b,a c,a d,b d,b e,c f,d c,d e,d f,e f/: x,kapacita,naklady;
endsets

data:
kapacita = 2 3 4 1 4 4 5 9 8 3;
naklady = 5 4 6 6 7 8 2 1 6 2;
T = 9;
enddata

min = @sum(tok: x*naklady);
@sum(tok(i,j)|i#EQ#1: x(i,j))>=T;
@for(uzel(i)|i#NE#1#AND#i#NE#6: @sum(tok(i,j): x(i,j)) = @sum(tok(j,i): x(j,i)));
@for(tok: x<=kapacita);
end
```

Zadání



Řešení




```
Minimalni naklady jsou 109.
```

MAXIMALIZACE TOKU PŘI ZADANÝCH NÁKLADECH

Máme graf s jedním počátečním uzlem (zdroj) a jedním koncovým uzlem (místo určení). Známe kapacitu hran k_{ij} mezi každou dvojicí uzlů *i*, *j*. Zároveň známe náklady c_{ij} spojené s jednotkou toku hranou *i*, *j*. Naším cílem je maximalizovat tok grafem, ale nepřekročit přitom povolené náklady C_0.

Zavedeme jednu proměnnou x_{ij}, která označuje objem toku z uzlu *i* do uzlu *j*.

$z = \sum_{j_{(1,j)\epsilon H}} x_{1j} \rightarrow max$	*Snažíme se maximalizovat buď celkový tok z prvního uzlu*
$\sum_{j_{(i,j)\epsilon H}} x_{ij} - \sum_{k_{(k,i)\epsilon H}} x_{ki} = 0 \quad i \neq 1, n$	*(1) Pro každý uzel kromě prvního a posledního musí platit, že to, co do i-tého uzlu celkem vteče, se musí rovnat tomu, co z něj celkem vyteče. Sčítáme přes všechny hrany, které vedou z daného uzlu a do něj.*
$0 \leq x_{ij} \leq k_{ij} \quad (i,j)\epsilon H$	*(2) Pro každou hranu mezi dvojicí uzlů i, j musí platit, že jí neproteče více, než kolik je její kapacita.*
$\sum_i \sum_{j_{(i,j)\epsilon H}} c_{ij} x_{ij} \leq C_0$	*(3) Celkové náklady na tok nesmí překročit povolené náklady.*

MAXIMALIZACE TOKU S FIXNÍMI NÁKLADY

Máme graf s jedním počátečním uzlem (zdroj) a jedním koncovým uzlem (místo určení). Známe kapacitu hran k_{ij} mezi každou dvojicí uzlů i, j. Zároveň známe fixní (například investiční) náklady c_{ij} na hranu mezi uzly *i*, *j*. To znamená, že pokud tuto hranu použijeme, zaplatíme c_{ij} bez ohledu na objem toku touto hranou. Naším cílem je zajistit požadovanou hodnotu celkového toku T_0 s co nejnižšími náklady.

Zavedeme dvě proměnné:

- proměnná x_{ij}, která označuje objem toku z uzlu *i* do uzlu *j*.
- binární proměnná y_{ij}, která bude rovna 1 v případě, že bude hrana mezi uzly *i*, *j* použita

$z = \sum_i \sum_{j_{(i,j)\epsilon H}} c_{ij} y_{ij} \rightarrow min$	*Snažíme se minimalizovat celkové náklady spojené s tokem.*
$\sum_{j_{(i,j)\epsilon H}} x_{ij} - \sum_{k_{(k,i)\epsilon H}} x_{ki} = 0 \quad i \neq 1, n$	*(1) Pro každý uzel kromě prvního a posledního musí platit, že to, co do i-tého uzlu celkem vteče, se musí rovnat tomu, co z něj celkem vyteče. Sčítáme přes všechny hrany, které vedou z daného uzlu a do něj.*
$0 \leq x_{ij} \leq k_{ij} y_{ij} \quad (i,j)\epsilon H$	*(2) Pro každou hranu mezi dvojicí uzlů i, j musí platit, že pokud bude daná hrana použita, a tedy pokud bude y_{ij} rovno 1, neproteče jí více, než kolik je její kapacita. V opačném případě jí neproteče nic, poněvadž pravá strana bude rovna 0.*
$\sum_{j_{(1,j)\epsilon H}} x_{1j} \geq T_0$	*(3) To, co dohromady vyteče z prvního uzlu (tedy zároveň i celkový průtok grafem) musí být větší nebo rovno požadované hodnotě celkového toku,*
$y_{ij} \in \{0, 1\} \quad (i,j)\epsilon H$	*(4) Bude hrana mezi uzly i,j použita?*

Pokud by v předchozích modelech existovalo kapacitní omezení **uzlů** (tzn. může jimi protéct jen určitý objem d_j), pak přidáme podmínku:

$\sum_{i_{(i,j)\epsilon H}} x_{ij} \leq d_j \quad j \neq 1, n$	*Pro každý z j uzlů kromě prvního a posledního musí platit, že suma všeho, co do něj přiteče, nepřekročí jeho kapacitní omezení.*

VÍCEPRODUKTOVÉ TOKOVÉ ÚLOHY

U víceproduktových úloh protékají grafem různé produkty 1,2…*K*, takže proměnné mají tři indexy. Můžeme se například snažit minimalizovat celkové náklady spojené s tokem všech produktů. Známe náklady spojené s jednotkou toku produktu *k* hranou *ij:* c_{ij}^k. Také známe požadovaný tok produktu *k*: T_0^k

Proměnná x_{ij}^k říká, jaký objem *k*-tého produktu protéká hranou mezi uzly *i, j*.

$z = \sum_i \sum_{j_{(i,j)\epsilon H}} \sum_{k=1}^{K} c_{ij}{}^k x_{ij}{}^k \to min$	*Snažíme se minimalizovat celkové náklady spojené s tokem všech produktů.*
$\sum_{j_{(i,j)\epsilon H}} x_{ij}{}^k - \sum_{l_{(l,i)\epsilon H}} x_{li}{}^k = 0 \quad \forall k, i \neq 1, n$	*(1) Pro každý uzel kromě prvního a posledního musí platit, že objem k-tého produktu, který do i-tého uzlu vteče, se musí rovnat objemu k-tého produktu, který z něj vyteče. Sčítáme přes všechny hrany, které vedou z daného uzlu a do něj. Musí to platit pro každý z produktů.*
$\sum_{k=1}^{K} x_{ij}{}^k \leq k_{ij} \quad (i,j)\,\epsilon H$	*(2) Pro každou hranu mezi dvojicí uzlů i, j musí platit, že jí neproteče více, než kolik je její kapacita.*
$\sum_{j_{(1,j)\epsilon H}} x_{1j}{}^k \geq T_0{}^k \quad k = 1,2\ldots K$	*(3) Celkový objem k-tého produktu, který dohromady vyteče z prvního uzlu (tedy zároveň i celkový průtok k-tého produktu grafem) musí být větší nebo roven požadované hodnotě celkového toku k-tého produktu.*
$x_{ij}{}^k \geq 0 \quad k = 1,2\ldots K, (i,j)\,\epsilon H$	*(4) Tok bude nezáporný pro každou hranu a produkt.*

PŘEPRAVNÍ PROBLÉM (TRANSSHIPMENT PROBLEM)

Uvažujme situaci, kdy u každého uzlu máme zadán parametr a_i. Pokud je kladný, je tento uzel dodavatelem (zdrojem), pokud je záporný, je odběratelem (místem určení), pokud je roven 0, jde o průběžný uzel (překladiště). Dále máme opět zadánu kapacitu hrany (*i,j*) k_{ij} a náklady spojené s jednotkou toku c_{ij}. Cílem je splnit požadavky odběratelů a nepřekročit kapacity dodavatelů při minimálních nákladech spojených s tokem.

Zavedeme jednu proměnnou x_{ij}, která označuje objem toku mezi uzly *i, j*.

$z = \sum_i \sum_{j_{(i,j)\epsilon H}} c_{ij} x_{ij} \to min$	*Snažíme se minimalizovat celkové náklady spojené s tokem.*
$\sum_{j_{(i,j)\epsilon H}} x_{ij} - \sum_{k_{(k,i)\epsilon H}} x_{ki} = a_i \; \forall i$	*(1) Pro každý uzel musí platit, že to, co do i-tého uzlu vteče, mínus to, co z něj vyteče, se rovná* ***objemu zdrojů či požadavků daného uzlu.*** *Sčítáme přes všechny hrany, které vedou z daného uzlu a do něj.*
$0 \leq x_{ij} \leq k_{ij} \quad (i,j)\,\epsilon H$	*(2) Pro každou hranu mezi dvojicí uzlů i, j musí platit, že jí neproteče více, než kolik je její kapacita.*

!Minimalizujte celkové náklady spojené s tokem orientovaným grafem, který má 6 uzlů označených a, b, c, d, e, f. Spojení mezi uzly, příslušné kapacity hran a náklady spojené s jednotkou toku ranou jsou uvedené v tabulce:

Spojení	**Kapacita**	**Náklady**	**Spojení**	**Kapacita**	**Náklady**
a,b	10	5	c,e	7	6
a,c	10	10	c,f	5	9
a,d	12	20	d,c	3	12
b,e	11	11	d,f	9	17
c,d	3	12	e,f	18	8

```
Množství produktu, které charakterizuje uzel je: a = 15, b = 0, c = 10, d =
-5, e = 0, f = -20;
model:
sets:
uzel/a,b,c,d,e,f/:produkt;
hrana(uzel,uzel)/a b,a c,a d,b e,c d,c e,c f,d c,d f,e f/:
kapacita,x,naklady;
endsets

data:
kapacita=10 10 12 11 3 7 5 3 9 18;
naklady= 5, 10, 20, 11, 12, 6, 9, 12, 17, 8;
produkt=15 0 10 -5 0 -20;
enddata

min = @sum(hrana: naklady*x);
@for(hrana: x<=kapacita);
@for(uzel(i): @sum(hrana(i,j): x(i,j)) - @sum(hrana(k,i): x(k,i)) =
produkt(i));!odtok mínus přítok odpovídá požadavkům nebo kapacitám uzlů;
end
Zdroj: Ing. Jan Fábry, Ph.D.: 4EK314 Diskrétní modely
```

Zadání



Řešení



MINIMÁLNÍ KOSTRA (MINIMAL SPANNING TREE)

Uvažujme graf s množinou uzlů V a množinou hran H. Cílem je najít kostru tohoto grafu s minimálním ohodnocením hran, tzn. G' = {V,H'}, H' ϵ H. Známe přitom náklady spojené s nákladem na zřízení (využití) jednotlivých hran.

Úlohu převádíme na transshipment problem. Zavedeme dvě proměnné:

- binární proměnná x_{ij} se rovná 1 v případě, že je hrana (i,j) vybrána, jinak 0
- proměnná y_{ij} je tok hranou (i,j)

Princip řešení si nejdříve ukážeme na obrázku (model je níže vyřešen v Lingu).

Uzel číslo 1 je uzlem, kam vše poteče, ale z něj nic nevytéká. Je tedy „odběratelem," odebere právě *n* – 1 hran, protože právě tolik hran musí výsledný strom obsahovat, aby bylo všech *n* uzlů propojeno.

Každý ze zbylých uzlů je „dodavatelem" s kapacitou 1, protože „přidá" do stromu jednu hranu.

V následujícím stromu budou tedy proměnné x_{52}, x_{21}, x_{31}, x_{43} rovny 1, protože tyto hrany budou využity. Zbylé proměnné x budou rovny 0.

Z proměnných y_{ij} *budou nenulové jen* y_{52}, y_{21}, y_{31}, y_{43}. Proměnná y_{43} se bude rovnat jedné, protože do uzlu 4 nic nevteklo, ale jedna hrana z něj musí vytéct, jako koneckonců z každého uzlu kromě prvního. Proměnná y_{31} se bude rovnat dvěma, protože do uzlu přitekl tok o objemu jedna (1 hrana ze čtvrtého uzlu), což z něj musí zase vytéct, ale zároveň se k tomu musí přidat další hrana. Analogicky se bude proměnná y_{52} rovnat jedné a proměnná y_{21} dvěma. Celkem do prvního uzlu přijde tok o objemu 4, což odpovídá čtyřem hranám.





$z = \sum_i \sum_{j_{(i,j)\epsilon H}} c_{ij} x_{ij} \rightarrow min$	*Snažíme se minimalizovat celkové náklady spojené s tokem. Náklady platíme pouze za hrany (i,j), které vybereme do stromu.*
$\sum_{j_{(i,j)\epsilon H}} y_{ij} - \sum_{k_{(k,i)\epsilon H}} y_{ki} = 1 \quad i \in V - \{1\}$	*(1) Pro každý uzel musí platit, že počet hran, které z něj vytečou, mínus počet hran, které do něj vtečou, se rovná 1. Jinak řečeno, každý uzel přidá do stromu právě jednu hranu. Výjimkou je první uzel, kam vše jen vtéká, ale nic z něj nevytéká.*
$0 \leq y_{ij} \leq (n-1) x_{ij} \quad (i,j)\,\epsilon H$	*(2) Pro každou hranu (i,j) musí platit, že pokud bude vybrána, proteče jí nejvýše tok o objemu n – 1, a pokud vybrána nebude, neproteče jí nic.*
$x_{1j} = 0 \quad j \in V - \{1\}$	*(3) Z prvního uzlu žádná hrana nevede. Je de facto koncovým uzlem, do něhož hrany směřují.*
$\sum_{j_{(i,j)\epsilon H}} x_{ij} = 1 \quad i \in V - \{1\}$	*(4) Z každého uzlu kromě prvního povede právě jedna hrana.*
$x_{ij} \in \{0,1\} \quad (i,j)\,\epsilon H$	*(5) Bude hrana (i,j) použita?*

```
!Najděte minimální kostru grafu s pěti uzly, kde náklady na zřízení hran
jsou:
 C = 0 2 3 4 5
        2 0 4 5 2
        3 4 0 2 3
        4 5 2 0 4
        5 2 3 4 0.
;

model:
sets:
uzel/1..5/;
tok(uzel,uzel): x,y,C; !x = bude hrana vybrána? y = kolik poteče;
endsets

data:
C =
0 2 3 4 5
2 0 4 5 2
3 4 0 2 3
4 5 2 0 4
5 2 3 4 0;
enddata

min = @sum(tok: x*C);
@for(tok: @bin(x));
@for(uzel(i)|i#NE#1: @sum(uzel(j): x(i,j)) = 1);!z každého kromě 1.uzlu
poteče právě 1 hrana;
@for(tok: y<=4*x);
@for(uzel(i)|i#NE#1: @sum(uzel(j): y(i,j)) - @sum(uzel(k): y(k,i)) =
1);!každý uzel přidá do toku právě jednu hranu;
@for(uzel(i)|i#EQ#1: @sum(uzel(j): x(i,j)) = 0);
end
Zdroj: Ing. Jan Fábry, Ph.D.: 4EK314 Diskrétní modely
```

MINIMÁLNÍ STEINERŮV STROM

Cílem je podobně jako u minimální kostry najít co nejlevnější podgraf, abychom zajistili propojení uzlů. Rozdíl spočívá v tom, že je zde ale možnost zapojit tzv. průběžné uzly (překladiště), pokud bude díky nim propojení levnější. Snažíme se tedy propojit všechny koncové uzly se zdrojem (přes průběžné uzly nebo přímo) tak, aby náklady byly minimální.

Pracujeme se třemi typy uzlů: zdroj (to označíme jako uzel 1), koncové uzly (ty označíme jako uzel 2, 3, …p) překladiště (průběžné uzly, pomocné stanice, ústředny, označíme jako uzly p+1,p+2…n). Přitom známe jednak náklady c_{ij} na hranu (*i,j*), jednak náklady na zřízení ústředny (d_j).

Zavedeme tři proměnné:

- binární proměnná x_{ij} se rovná 1 v případě, že je hrana (i,j) vybrána, jinak 0
- binární proměnná f_i se rovná 1 v případě, že je uzel i zahrnutý, jinak 0; to znamená, že pro koncové uzly a zdroj ji musíme pevně nastavit na 1, zatímco u průběžných uzlů může nabývat hodnot 0, 1;
- proměnná y_{ij} je tok hranou (i,j)

$z = \sum_{i=1}^{n}\sum_{j=1}^{n} c_{ij}x_{ij} + \sum_{i=1}^{n} d_i f_i \rightarrow min$	*Snažíme se minimalizovat celkové náklady spojené s tokem. Náklady platíme pouze za hrany (i,j), které vybereme do stromu* ($x_{ij} = 1$). *K tomu navíc platíme náklady za každé „překladiště" které chceme také zahrnout* ($f_i = 1$).
$\sum_{j=1}^{n} y_{ij} - \sum_{j=2}^{n} y_{ji} = f_i \quad i = 2,3 \dots n$	*(1) Pro každý uzel musí platit, že počet hran, které z něj vytečou, mínus počet hran, které do něj vtečou, se rovná jedné, pokud bude tento uzel zahrnutý, a nule v opačném případě. Výjimkou je první uzel, kam vše jen vtéká, ale nic z něj nevytéká.*
$0 \leq y_{ij} \leq (n-1)x_{ij} \quad i,j = 1,2 \dots n$	*(2) Pro každou hranu (i,j) musí platit, že pokud bude vybrána, proteče jí nejvýše tok o objemu n – 1, a pokud vybrána nebude, neproteče jí nic.*
$x_{1j} = 0 \quad j = 2,3 \dots n$	*(3) Z prvního uzlu žádná hrana nevede. Je de facto koncovým uzlem, do něhož hrany směřují.*
$\sum_{j=1}^{n} x_{ij} = f_i \quad i = 2,3 \dots n$	*(4) Z každého uzlu kromě prvního povede právě jedna hrana, pokud bude tento uzel zahrnutý, a nula hran v opačném případě.*
$x_{ij} \in \{0,1\} \quad i,j = 1,2 \dots n$	*(5) Bude hrana (i,j) použita?*
$f_i \in \{0,1\} \quad i,j = p+1, p+2 \dots n$	*(6) Bude překladiště zahrnuto?*
$f_i = 1 \quad i,j = 1,2 \dots p$	*(7) Koncové zdroje a zdroj musí být zahrnuty.*

```
!Optimalizujte projekt kabelových rozvodů mezi třemi uživateli této
kabelové sítě. Kromě uživatelů (kteří tvoří jednu skupinu uzlů) máme i dvě
pomocné stanice, které však není nutné zahrnout do optimálního projektu a
jedna ústředna.
Náklady na zřízení rozvodů jsou: 15, 3, 18, 4, 7, 9, 6, 7, 12 (2-1, 2-5, 3-
1, 3-5, 3-6, 4-5, 4-6, 5-1, 6-1) a  náklady na zřízení uzlu jsou: 0, 0, 0,
0, 30, 20. Minimalizujte celkové náklady.
(pozn.: f1,2,3,4 = 1);

model:
sets:
uzel/1..6/:f,nakladyuzel;
tok(uzel,uzel)/2 1,2 5,3 1,3 5,3 6,4 5,4 6,5 1,6 1/:x,y,nakladytok;
endsets

data:
nakladyuzel=0 0 0 0 30 20;
```

```
nakladytok=15 3 18 4 7 9 6 7 12;
enddata

min = @sum(tok: x*nakladytok) + @sum(uzel: f*nakladyuzel);
@for(uzel(i)|i#GE#5: @bin(f(i)));!překladiště nemusí být zapojena;
@for(uzel(i)|i#LT#5: f(i) = 1);!první čtyři uzly jsou zapojeny;
@for(uzel(i)|i#NE#1: @sum(tok(i,j): x(i,j)) = f(i));!suma cest odpovídá 1,
je-li uzel zapojen, jinak 0;
@for(uzel(i)|i#NE#1: @sum(tok(i,j): y(i,j)) - @sum(tok(j,i)|j#GT#1: y(j,i))
= f(i));!suma výtoku mínus suma přítoku odpovídá 1, je-li uzel zapojen,
jinak 0;
@for(tok: y<=5*x);!nic neteče, není-li hrana zapojena;
@for(tok: @bin(x));
end
```





Zdroj: Ing. Jan Fábry, Ph.D.: 4EK314 Diskrétní modely

ZDROJE:

Ing. J. Fábry, Ph.D.: přednášky 4EK314 Diskrétní modely, 2011.

Teorie grafů, VŠB 2008. Dostupné z: http://books.fs.vsb.cz/SystAnal/texty/21.htm